Sarjaratkaisun etsiminen Maplella

Olkoon tarkasteltavana ensimmäisen kertaluvun differentiaaliyhtälö:

```
> diffyht:= diff(y(x), x)=1+y(x)^2;
```

$$diffyht := \frac{d}{dx}\,\mathrm{y}(x) = 1 + \mathrm{y}(x)^2$$

Tälle pyritään etsimään sarjaratkaisu origokeskisenä potenssisarjana. Tavoitteena on laskea sarjan termien kertoimet *n* -asteiseen termiin saakka:

```
> n:= 10;
```

$$n := 10$$

Potenssisarjojen käsittelyä varten on *Maple*ssa paketti **`powseries`**.

```
> with(powseries):
```

Tarvittava potenssisarjamuotoinen yrite on

```
> powcreate(yrite(k)=a[k]):
  tpsform(yrite, x, n);
```

$$a_0 + a_1\,x + a_2\,x^2 + a_3\,x^3 + a_4\,x^4 + a_5\,x^5 + a_6\,x^6 + a_7\,x^7 + a_8\,x^8 + a_9\,x^9 + \mathrm{O}(x^{10})$$

`Powcreate`-komennolla luodaan *Maple*-proseduuri, joka vastaa haluttua potenssisarjaa. **`Tpsform`**-komennolla saadaan tulostettua haluttu määrä termejä luodusta potenssisarjasta. **`Powseries`**-pakettia käytettäessä on potenssisarjojen väliset operaatiot tehtävä paketin työkaluilla. Esimerkiksi kahta **`powcreate`**-komennolla luotua potenssisarjaa ei voida kertoa keskenään operaattorilla *, vaan on käytettävä paketin komentoa **`multiply`**. Vastaavasti **`diff`** -komento ei pure potenssisarjaan vaan on käytettävä komentoa **`powdiff`**.

Sijoittamista varten on paketin työkaluilla laskettava valmiiksi alkuperäisessä differentiaaliyhtälössä esiintyvät derivointi ja toiseen potenssiin korottaminen. Tämän jälkeen potenssisarjoista muodostetaan sijoitusta varten polynomit ottamalla mukaan *n* ensimmäistä termiä.

```
> yriteder:= powdiff(yrite):
  tpsform(yriteder, x, n);
```

$$a_1 + 2\,a_2\,x + 3\,a_3\,x^2 + 4\,a_4\,x^3 + 5\,a_5\,x^4 + 6\,a_6\,x^5 + 7\,a_7\,x^6 + 8\,a_8\,x^7 + 9\,a_9\,x^8 + 10\,a_{10}\,x^9 + \mathrm{O}(x^{10})$$

```
> yritetoiseen:= multiply(yrite, yrite):
  tpsform(yritetoiseen, x, n);
```

$$a_0^2 + 2\,a_0\,a_1\,x + (2\,a_0\,a_2 + a_1^2)\,x^2 + (2\,a_0\,a_3 + 2\,a_1\,a_2)\,x^3 + (2\,a_0\,a_4 + 2\,a_1\,a_3 + a_2^2)\,x^4 +$$

$(2\,a_0\,a_5+2\,a_1\,a_4+2\,a_2\,a_3)\,x^5+(2\,a_0\,a_6+2\,a_1\,a_5+2\,a_2\,a_4+a_3^{\;2})\,x^6+$

$(2\,a_0\,a_7+2\,a_1\,a_6+2\,a_2\,a_5+2\,a_3\,a_4)\,x^7+(2\,a_0\,a_8+2\,a_1\,a_7+2\,a_2\,a_6+2\,a_3\,a_5+a_4^{\;2})\,x^8+$

$(2\,a_0\,a_9+2\,a_1\,a_8+2\,a_2\,a_7+2\,a_3\,a_6+2\,a_4\,a_5)\,x^9+\mathrm{O}(x^{10})$

```
> yritederpol:= convert(tpsform(yriteder, x, n), polynom);
```

$$yritederpol := a_1+2\,a_2\,x+3\,a_3\,x^2+4\,a_4\,x^3+5\,a_5\,x^4+6\,a_6\,x^5+7\,a_7\,x^6+8\,a_8\,x^7+9\,a_9\,x^8+10\,a_{10}\,x^9$$

```
> yritetoiseenpol:= convert(tpsform(yritetoiseen, x, n), polynom);
```

$$yritetoiseenpol := a_0^{\;2}+2\,a_0\,a_1\,x+(2\,a_0\,a_2+a_1^{\;2})\,x^2+(2\,a_0\,a_3+2\,a_1\,a_2)\,x^3$$
$$+(2\,a_0\,a_4+2\,a_1\,a_3+a_2^{\;2})\,x^4+(2\,a_0\,a_5+2\,a_1\,a_4+2\,a_2\,a_3)\,x^5$$
$$+(2\,a_0\,a_6+2\,a_1\,a_5+2\,a_2\,a_4+a_3^{\;2})\,x^6+(2\,a_0\,a_7+2\,a_1\,a_6+2\,a_2\,a_5+2\,a_3\,a_4)\,x^7$$
$$+(2\,a_0\,a_8+2\,a_1\,a_7+2\,a_2\,a_6+2\,a_3\,a_5+a_4^{\;2})\,x^8$$
$$+(2\,a_0\,a_9+2\,a_1\,a_8+2\,a_2\,a_7+2\,a_3\,a_6+2\,a_4\,a_5)\,x^9$$

Sijoitetaan yrite differentiaaliyhtälöön.

```
> sarjayht:= yritederpol=1+yritetoiseenpol;
```

$$sarjayht := a_1+2\,a_2\,x+3\,a_3\,x^2+4\,a_4\,x^3+5\,a_5\,x^4+6\,a_6\,x^5+7\,a_7\,x^6+8\,a_8\,x^7+9\,a_9\,x^8+10\,a_{10}\,x^9=1$$
$$+a_0^{\;2}+2\,a_0\,a_1\,x+(2\,a_0\,a_2+a_1^{\;2})\,x^2+(2\,a_0\,a_3+2\,a_1\,a_2)\,x^3+(2\,a_0\,a_4+2\,a_1\,a_3+a_2^{\;2})\,x^4$$
$$+(2\,a_0\,a_5+2\,a_1\,a_4+2\,a_2\,a_3)\,x^5+(2\,a_0\,a_6+2\,a_1\,a_5+2\,a_2\,a_4+a_3^{\;2})\,x^6$$
$$+(2\,a_0\,a_7+2\,a_1\,a_6+2\,a_2\,a_5+2\,a_3\,a_4)\,x^7+(2\,a_0\,a_8+2\,a_1\,a_7+2\,a_2\,a_6+2\,a_3\,a_5+a_4^{\;2})\,x^8$$
$$+(2\,a_0\,a_9+2\,a_1\,a_8+2\,a_2\,a_7+2\,a_3\,a_6+2\,a_4\,a_5)\,x^9$$

Yhtälön oikealla ja vasemmalla puolella on potenssisarja. Jotta nämä olisivat samat, tulee samakorkuisten potenssien kertoimien yhtälön oikealla ja vasemmalla puolella olla samat. Vähentämällä yhtälön oikea puoli vasemmasta saadaan aikaan polynomi, jonka potenssien kertoimien tulee tällöin olla 0. Poimimalla kertoimet **`coeffs`**-komennolla ja määrittelemällä ne nollaksi, saadaan yhtälöryhmä.

```
> lhs(sarjayht)-rhs(sarjayht):
  coeffs(%, x):
  yhtalot:= {seq(%[i]=0, i=1..n)};
```

$$yhtalot := \{a_1-1-a_0^{\;2}=0,\ 2\,a_2-2\,a_0\,a_1=0,\ -2\,a_0\,a_4-2\,a_1\,a_3-a_2^{\;2}+5\,a_5=0,$$
$$6\,a_6-2\,a_0\,a_5-2\,a_1\,a_4-2\,a_2\,a_3=0,\ -2\,a_0\,a_6-2\,a_1\,a_5-2\,a_2\,a_4-a_3^{\;2}+7\,a_7=0,$$
$$8\,a_8-2\,a_0\,a_7-2\,a_1\,a_6-2\,a_2\,a_5-2\,a_3\,a_4=0,$$
$$9\,a_9-2\,a_0\,a_8-2\,a_1\,a_7-2\,a_2\,a_6-2\,a_3\,a_5-a_4^{\;2}=0,$$

$$10\,a_{10}-2\,a_0\,a_9-2\,a_1\,a_8-2\,a_2\,a_7-2\,a_3\,a_6-2\,a_4\,a_5=0,\ -2\,a_0\,a_3-2\,a_1\,a_2+4\,a_4=0,$$
$$-2\,a_0\,a_2-a_1^{\,2}+3\,a_3=0\,\}$$

Tuloksena on rekursiivinen epälineaarinen yhtälöryhmä: ensimmäisestä yhtälöstä voidaan ratkaista a_1, jos a_0 tunnetaan, toisesta tämän jälkeen a_2, kolmannesta a_3 jne. Kokonaisuudessaan ryhmä voidaan ratkaista **solve**-komennolla. Tässä jälkimmäiseksi argumentiksi pitäisi oikeastaan antaa lista yhtälöryhmän tuntemattomista, mutta oletuksena on, että ratkaisu tapahtuu kaikkien ryhmässä esiintyvien symbolien suhteen. Kaikkia tuntemattomia ei saada ratkaistuiksi, vaan muut voidaan ainoastaan lausua ensimmäisen kertoimen a_0 avulla.

```
> ratk:= solve(yhtalot);
```

$$ratk := \Big\{a_3=\frac{4}{3}a_0^{\,2}+a_0^{\,4}+\frac{1}{3},\ a_4=\frac{5}{3}a_0^{\,3}+a_0^{\,5}+\frac{2}{3}a_0,\ a_5=2\,a_0^{\,4}+a_0^{\,6}+\frac{17}{15}a_0^{\,2}+\frac{2}{15},$$
$$a_7=\frac{8}{3}a_0^{\,6}+a_0^{\,8}+\frac{12}{5}a_0^{\,4}+\frac{248}{315}a_0^{\,2}+\frac{17}{315},\ a_6=\frac{7}{3}a_0^{\,5}+a_0^{\,7}+\frac{77}{45}a_0^{\,3}+\frac{17}{45}a_0,$$
$$a_9=\frac{37}{9}a_0^{\,6}+\frac{10}{3}a_0^{\,8}+\frac{424}{189}a_0^{\,4}+\frac{1382}{2835}a_0^{\,2}+a_0^{\,10}+\frac{62}{2835},\ a_2=a_0\,(1+a_0^{\,2}),$$
$$a_{10}=\frac{14102}{14175}a_0^{\,3}+\frac{3179}{945}a_0^{\,5}+\frac{77}{15}a_0^{\,7}+a_0^{\,11}+\frac{1382}{14175}a_0+\frac{11}{3}a_0^{\,9},$$
$$a_8=3\,a_0^{\,7}+a_0^{\,9}+\frac{16}{5}a_0^{\,5}+\frac{88}{63}a_0^{\,3}+\frac{62}{315}a_0,\ a_0=a_0,\ a_1=1+a_0^{\,2}\Big\}$$

Sijoittamalla kertoimet yritteeseen saadaan ratkaisuna olevan potenssisarjan alkupää:

```
> sarjaratk:= subs(ratk, tpsform(yrite, x, n+1));
```

$$sarjaratk := a_0+(1+a_0^{\,2})\,x+a_0\,(1+a_0^{\,2})\,x^2+\left(\frac{4}{3}a_0^{\,2}+a_0^{\,4}+\frac{1}{3}\right)x^3+\left(\frac{5}{3}a_0^{\,3}+a_0^{\,5}+\frac{2}{3}a_0\right)x^4+$$
$$\left(2\,a_0^{\,4}+a_0^{\,6}+\frac{17}{15}a_0^{\,2}+\frac{2}{15}\right)x^5+\left(\frac{7}{3}a_0^{\,5}+a_0^{\,7}+\frac{77}{45}a_0^{\,3}+\frac{17}{45}a_0\right)x^6+$$
$$\left(\frac{8}{3}a_0^{\,6}+a_0^{\,8}+\frac{12}{5}a_0^{\,4}+\frac{248}{315}a_0^{\,2}+\frac{17}{315}\right)x^7+\left(3\,a_0^{\,7}+a_0^{\,9}+\frac{16}{5}a_0^{\,5}+\frac{88}{63}a_0^{\,3}+\frac{62}{315}a_0\right)x^8+$$
$$\left(\frac{37}{9}a_0^{\,6}+\frac{10}{3}a_0^{\,8}+\frac{424}{189}a_0^{\,4}+\frac{1382}{2835}a_0^{\,2}+a_0^{\,10}+\frac{62}{2835}\right)x^9+$$
$$\left(\frac{14102}{14175}a_0^{\,3}+\frac{3179}{945}a_0^{\,5}+\frac{77}{15}a_0^{\,7}+a_0^{\,11}+\frac{1382}{14175}a_0+\frac{11}{3}a_0^{\,9}\right)x^{10}+\mathrm{O}(x^{11})$$

Tämä sisältää yhden määräämättömän vakion a_0, kuten ensimmäisen kertaluvun differentiaaliyhtälön yleiselle ratkaisulle luonnollista onkin. Jos alkuehdoksi valitaan $y(0)=0$, tulee olla $a_0=0$. Vastaava yksittäisratkaisu on

```
> yksratk:= subs(a[0]=0, sarjaratk);
```

$$yksratk := x + \frac{1}{3}x^3 + \frac{2}{15}x^5 + \frac{17}{315}x^7 + \frac{62}{2835}x^9 + O(x^{11})$$

Tämä on sama kuin funktion tan(*x*) origokeskinen Taylorin sarja:

```
> series(tan(x), x, n+1);
```

$$x + \frac{1}{3}x^3 + \frac{2}{15}x^5 + \frac{17}{315}x^7 + \frac{62}{2835}x^9 + O(x^{11})$$

Näin tulee ollakin, sillä differentiaaliyhtälö on separoituva ja sen yleiseksi ratkaisuksi saadaan $y = \tan(x + C)$.

```
> dsolve(diffyht, y(x));
```

$$y(x) = \tan(x + \_C1)$$

Jotta sarjaratkaisulle voitaisiin piirtää kuvaaja, siitä on pudotettava jäännöstermi pois. Tämän tarkkaa lauseketta ei tunneta eikä sille siis voida laskea numeerisia arvoja piirtämistä varten. Jäännöstermin poistaminen tapahtuu muuntamalla potenssisarja polynomiksi komennolla **`convert`**.

```
> poly:= convert(yksratk, polynom);
```

$$poly := x + \frac{1}{3}x^3 + \frac{2}{15}x^5 + \frac{17}{315}x^7 + \frac{62}{2835}x^9$$

Sarjaratkaisu ja funktio tan(*x*) samassa kuvassa:

```
> plot({poly, tan(x)}, x=0..1.5);
```



Esitetty lasku ei anna viitteitä sarjaratkaisun suppenemisalueesta. Kokonaan muilla keinoilla voidaan osoittaa, että tangentin origokeskisen Taylorin sarjan suppenemissäde on $\frac{\pi}{2}$. Sarja suppenee siis vain välillä $]-\frac{\pi}{2}, \frac{\pi}{2}[$.

Haluttua termilukua voidaan edellä olevassa laskussa muuttaa ja tämän jälkeen laskea kaikki uudelleen valinnalla Evaluate Notebook valikosta Kernel/Evaluation.

Ratkaisu suoraan dsolve-komennolla

*Maple*n **`dsolve`**-komennolla on mahdollista laskea suoraan ratkaisuja, jotka ovat Taylorin sarjoja. Tällöin käytetään parametria **`type=series`**.

Esimerkkinä edellä johdettu sarjaratkaisu suoraan **`dsolve`**-komennolla:

```
> Order:=10:
  dsolve({diffyht, y(0)=0}, y(x), type=series):
  simplify(%);
```

$$y(x) = x + \frac{1}{3}x^3 + \frac{2}{15}x^5 + \frac{17}{315}x^7 + \frac{62}{2835}x^9 + O(x^{10})$$

```
>
```

Linkit

sarjamuotoinen yrite

SKK & MS 31.05.2001